

Infinite Words and Morphic Languages Formalized in Isabelle/HOL

Štěpán Starosta²[0000–0001–5962–4297]

Czech Technical University in Prague, Czech Republic

Abstract. We present a formalization of basics related to infinite words in the generic proof assistant Isabelle/HOL. Furthermore, we present a formalization of purely morphic and morphic languages. Finally, we present a formalized definition of Sturmian words as lower mechanical words and prove some very elementary facts. The formalization is based on an ongoing larger project of formalization of combinatorics on words.

Keywords: infinite word · morphic language · Sturmian word · Isabelle/HOL · mathematics formalization.

1 Introduction

Infinite words and their various properties are a popular object of study in combinatorics on words. While the beginning of such studies is usually attributed to M. Prouhet’s work [18] of 1851 and A. Thue’s work [20] some 50 years later, the beginning of a systematic study and raise of the field is marked by the creation of the first of Lothaire’s book [14]. For a detailed description of the history and on the studied concepts one may refer to [3].

This article presents a formalization of elementary concepts related to infinite words from the point of view of combinatorics on words. The goals of mathematics formalization is to write down statements in a formal way and provide them with machine-verified proofs using a given set of inference rules. With the raise of computing power, this field has grown considerably since the first (large) computer-assisted proof of the four-color theorem in 1970s. The reader may refer to the survey [12] for a more detailed overview of the state of mathematics formalization. From a different angle, the evolution of this field and its available tools may be also tracked on the formalization state of 100 prominent theorems in various available provers [21].

The presented formalization is done in the generic proof assistant Isabelle/HOL [11,17]. It is an open-source project with a vast collection of formalized results available in the main distribution of Isabelle and in the Archive of Formal Proofs [1]. The interface language of Isabelle is called Isar, see [16]. One of Isar’s goals is to provide human-readable proofs. In what follows, we give many examples of Isar code copied as it is in the presented formalization, leaving to the reader to consider the achieved level of readability.

The formalization is within an existing project of formalization of combinatorics on words in Isabelle/HOL [5] called “Combinatorics on Words Formalized”. As of today, it covers many elementary and some advanced concepts from the field (related to finite words). Besides the most up-to-date repository [5], the reader may refer to [8,6,10] for more details on this project. In this context, the present article describes the extension of the project with the concepts of infinite words and, in what follows, gives highlights of important or interesting parts of this extension, omitting numerous other formalized claims and proofs.

In the next section we give a few more details on relevant concepts from the Combinatorics on Words Formalized project. The formalization is split into 3 parts, described Sections 3 to 5. The last section describes the structure of the formalization.

2 Preliminaries - combinatorics on finite words

The basic component of the Combinatorics on Words Formalized project, called “Combinatorics on Words Basics”, is also available in the Archive of Formal Proofs [9], where it is shortly called “Combinatorics_Words”. We shall refer to it by this name as it is the basic component which allows the formalization of the concepts presented in this article. For the same reason, in this section, we give a very brief overview of related notions available in Combinatorics_Words.

Finite words are finite sequences of elements from a given set \mathcal{A} , which is usually called an *alphabet*. The elements of an alphabet are *letters*. Finite words are naturally represented by the datatype of lists, denoted `'a list` in Isabelle, where `'a` is a type variable. Hence, `'a` represents the type for the letter, and all the elements of this type represent the alphabet. The empty list/word is `[]` or `Nil`, also denoted by ε in Combinatorics_Words.

The operation of concatenation is also provided by the “List” theory of Isabelle main distribution. A theory is the elementary unit of formalization in Isabelle and it coincides with a file having extension “thy”. Combinatorics_Words package defines the more usual notation of concatenation in combinatorics on words, namely, it is denoted in the formalization by the usual algebraic notation `.`. The length of a finite word is also defined in the List theory, and a usual notation $|w|$ ¹ is adopted in Combinatorics_Words. Similarly, the concepts of prefix and factor are already in the theory “HOL-Library.Sublist” which is reused by Combinatorics_Words. A frequent notation for these phenomena is introduced: $p \leq_p w$ stands for p being a prefix of w , and $f \leq_p w$ denotes that f is a factor of w . The mentioned notation, along with many elementary or slightly advanced related claims are in “CoWBasic” theory of Combinatorics_Words.

The concept of a morphism and endomorphism is introduced in “Morphisms” theory of Combinatorics_Words along with many useful claims. A morphism φ is a mapping from a set of finite words to another set of finite words such that

¹ The symbol “ $|$ ” is in boldface in the formalization to avoid conflict with existing notation.

for all w, v we have $\varphi(w \cdot v) = \varphi(w) \cdot \varphi(v)$ (it is a homomorphism from one word monoid to another).

3 Languages

The study of infinite words often comprises the study of their set of factors. Hence, naturally, before introducing infinite words, we introduce several notions relevant to languages, i.e., sets of finite words, in general.

A language in the formalization needs not be defined as it stems from the fact that the alphabet is represented by a type. In other words, a language is a variable of the type `'a list set`, i.e., set of variables of type `'a list`, that is, a set of finite words.

The first definition is a language closed under taking factors. Such language is called *factorial*: a language L is factorial if

$$\forall w \in L, \forall f, f \leq_f w \Rightarrow f \in L.$$

The formalization of this definition is straightforward:

```
definition factorial :: "'a list set  $\Rightarrow$  bool"
where "factorial L  $\equiv$  ( $\forall w f. w \in L \wedge f \leq_f w \longrightarrow f \in L$ )"
```

The type `"'a list set \Rightarrow bool"` represents a mapping from sets of `'a list`, i.e., sets of finite words, to `bool`, thus creating a predicate on a language. The symbol \longrightarrow stands for HOL implication, i.e., a binary operation acting on the `bool` type.

Taking the *factor closure* of a language L produces the smallest superset of L which is factorial. In the formalization, the command `inductive_set` is used to define the closure inductively:

```
inductive_set factor_closure :: "'a list set  $\Rightarrow$  'a list set" for L
where " $w \in L \Longrightarrow v \leq_f w \Longrightarrow v \in$  factor_closure L"
```

The long double arrow \Longrightarrow is Isabelle's metaimplication using to reason about claims. The last line should be read as if $w \in L$ and $v \leq_f w$, then v belongs to `factor_closure L`. The free variables `w` and `v` are universally quantified. The advantage of using this, very simple, inductive definition is that Isabelle produces many useful facts that can be used later in the formalization. For instance, the following induction rule, called `factor_closure.induct` is produced:

```
"x  $\in$  factor_closure L  $\Longrightarrow$ 
( $\bigwedge w v. w \in L \Longrightarrow v \leq_f w \Longrightarrow P v$ )  $\Longrightarrow P x$ "
```

This automatically produced rule may be read as if x is in the factor closure of a language L and for all w, v we have that

$$w \in L, v \leq_f w \text{ implies that } P(v) \text{ holds,}$$

then $P(x)$ holds.

There are more advantages of using the command `inductive_set` when used to define the following concept of a morphic language, which stems from the idea

of a D0L-system. A D0L-system is a specific instance of an L-system introduced in [13] to model growth of organisms and used nowadays to generate vegetations in virtual worlds. A classical overview of related concepts is in [19]. A D0L-system is a triple $(\mathcal{A}, \varphi, w)$ where \mathcal{A} is an alphabet, φ is an endomorphism over \mathcal{A}^* , the set of all finite words over \mathcal{A} , and w is a word over \mathcal{A} usually called *axiom*. The set $\{\varphi^k(w) \mid k = 0, 1, \dots\}$ is the language of the system. We are interested in the factorial closure of this language, that is, we are interested in any factor that can be produced by repeated application of φ on the axiom w . We call this language a *purely morphic language* and it is defined as follows:

```

inductive_set purely_morphic_language :: "('a list  $\Rightarrow$  'a list)
 $\Rightarrow$  'a list  $\Rightarrow$  'a list set"
  for f axiom where
    "axiom  $\in$  purely_morphic_language f axiom"
    | "w  $\in$  purely_morphic_language f axiom  $\Longrightarrow$  f w  $\in$ 
purely_morphic_language f axiom"
    | "w  $\in$  purely_morphic_language f axiom  $\Longrightarrow$  u  $\leq$  f w  $\Longrightarrow$  u  $\in$ 
purely_morphic_language f axiom"

```

There are two parameters in the above definition of `purely_morphic_language`. The first one, denoted `f`, is of type `'a list \Rightarrow 'a list`, which is of the same type as an endomorphism, but the requirement for it to actually be an endomorphism is not present. The reason is that the definition itself does not need this assumption, and it will be added when proving reasonable claims about this inductive set later. The second parameter is the axiom, denoted `axiom`.

There are 3 inductive rules:

1. the axiom is in the language;
2. the image by `f` of any element of the language is in the language;
3. a factor of any element of the language is in the language.

Considering the underlying datatype as the alphabet, it coincides with the factor closure of a language of a D0L-system for an endomorphism `f`. On the other hand, for a specific choice of axiom, it coincides with the set of all factors of a purely morphic infinite word, hence the name. An infinite word `u` *purely morphic* if there exists an endomorphism φ with $\varphi(\mathbf{u}) = \mathbf{u}$ and $\lim_{k \rightarrow +\infty} |\varphi^k(a)| = +\infty$ where a is the first letter of `u`. In other words $\varphi^k(a)$ constitute longer and longer prefixes of the word `u`.

The fixing of the parameter `f` to be an endomorphism is done by the command `context`, referring to the locale `endomorphism` from the required theory “Morphisms” of `Combinatorics_Words`. Locales are Isabelle’s tool to fix parameters to avoid their repetition in each claim. In this context, the endomorphism variable is fixed to `f`, and `\mathcal{L} w` is a shorthand for `purely_morphic_language f w`.

```

context endomorphism
begin

definition bounded :: "'a list  $\Rightarrow$  bool"
  where "bounded w  $\equiv$  (finite ( $\mathcal{L}$  w))"

```

The definition reads that a word w is *bounded* if the purely morphic languages induced by it is finite. The bounded words are interesting in the context when there are infinitely many bounded elements in a purely morphic language. Such systems are in [4] called *pushy*, which is the term we also adopt (f is still fixed to be an endomorphism):

```

definition pushy :: "'a list  $\Rightarrow$  bool"
  where "pushy axiom  $\equiv$   $\neg$  finite (( $\mathcal{L}$  axiom)  $\cap$  (Collect
bounded))"

```

As a highlight of a formalized elementary claim, we choose the following one stating that if an axiom is pushy with respect to the fixed endomorphism f if and only if it is pushy with respect to any non-trivial power of f :

```

lemma pushy_strong_shift:
  assumes "finite (UNIV::'a set)"
  shows "pushy axiom = endomorphism.pushy (fp + 1) axiom"

```

The assumption of the last lemma states that the universe of the datatype $'a$ representing the alphabet is finite.

While a proof is quite trivial for a non-erasing endomorphism f , it needs to be more elaborate when there is no such assumption. A good formalization forces the author not only to write down each claim correctly in the given formal language, but to choose (again and again) one of the many equivalent formulations which is the best for further reuse, as typically, most of the claims are reused later in further formalization. This choice comes with the revision of the assumptions which often ends in dropping or/and generalizing them. The price for dropping the assumption of f being non-erasing here is a somewhat more complicated proof that may be sketched as follows:

Proof (sketch of proof of lemma pushy_strong_shift). The implication from right to left is in fact easy. We show the converse, i.e., assume that **axiom** is pushy with respect to f .

Assume first that the maximum letter image length is greater than 1. Let $\Psi = f^{p+1}$. For contradiction assume that **axiom** is not pushy with respect to Ψ . It follows that there exists a constant ℓ such that any bounded factor of $\Psi^i(\mathbf{axiom})$ is shorter than ℓ . As **axiom** is pushy with respect to f , there exists a long enough bounded factor such that its preimage is some $\Psi^i(\mathbf{axiom})$ must be longer than ℓ . Here, long enough is based on the maximum letter image length of f . As a preimage of a bounded factor is bounded, we arrive at a contradiction.

The remaining case when the maximum letter image length is not greater than 1 is easy.

Note that the remaining assumption on the finiteness of the alphabet may not be dropped. Indeed, consider the following example tailored to make the claim fail. Assume $p > 0$, and consider the morphism φ determined by

$$b_i \mapsto \begin{cases} b_{i+1}^{i+1} & \text{if } i+1 \neq k(p+1); \\ \varepsilon & \text{otherwise,} \end{cases} \quad c_i \mapsto \begin{cases} c_{i+1} & \text{if } i \neq k(p+1); \\ b_{i+1}^{(i+1)!} c_{i+1} & \text{otherwise,} \end{cases}$$

The claim does not hold if the axiom equals c_0 since we obtain

$$\varphi^i(c_0) = \begin{cases} c_i & \text{if } i = k(p+1); \\ b_i^{i!} c_i & \text{otherwise.} \end{cases}$$

As $\varphi^{p+1}(b_i) = \varepsilon$, each letter b_i is bounded, hence the axiom c_0 is pushy with respect to φ . On the other hand, the letters c_i are not bounded (with respect to φ , which is equivalent² to be with respect to φ^{p+1}). As $(\varphi^{p+1})^i(c_0) = c_{i(p+1)}$, the axiom c_0 is not pushy with respect to φ^{p+1} .

A *morphic* word \mathbf{v} is a morphic image, by a morphism Ψ , of a purely morphic word, i.e., $\Psi(\varphi^k(a))$ constitute longer and longer prefixes of the word \mathbf{v} , provided that $\lim_{k \rightarrow +\infty} |\Psi(\varphi^k(a))| = +\infty$. We thus define a morphic language as follows:

```
definition morphic_language :: "('a list  $\Rightarrow$  'a list)  $\Rightarrow$  ('a list
 $\Rightarrow$  'b list)  $\Rightarrow$  'a list  $\Rightarrow$  'b list set"
where "morphic_language f h axiom  $\equiv$  factor_closure
(h'purely_morphic_language f axiom)"
```

The variable \mathbf{h} represents the outer morphism, and the language is simply the factor closure of the image by \mathbf{h} of `purely_morphic_language f axiom`. An inductive definition would also be possible, but that would not allow for a direct use of facts on the underlying purely morphic language. As a reuse of existing facts in formalization is rather preferred, the direct, less powerful, definition was chosen and specific useful claims are proven explicitly in the formalization. One such claim is the following induction rule with fixed endomorphism \mathbf{f} and morphism \mathbf{h} :

```
lemma morphic_language_induct[consumes 1]: assumes
  "x  $\in$  morphic_language f h axiom"
  "P (h axiom)" and
  ih2: " $\bigwedge w. h w \in$  morphic_language f h axiom  $\Rightarrow$  P (h w)  $\Rightarrow$ 
P (h (f w))" and
  ih3: " $\bigwedge w u. w \in$  morphic_language f h axiom  $\Rightarrow$  P w  $\Rightarrow$  u  $\leq$  f
w  $\Rightarrow$  P u"
shows "P x"
```

4 Infinite words

Similarly to the natural choice of the datatype for a finite word, an infinite word can be directly represented by the type `'a stream`, defined in the theory “HOL-

² This claim is formalized under the name `bounded_pow_bounded`.

Library.Stream” from the main distribution of Isabelle. In this theory, the type `'a stream` is defined as a (corecursive) codatatype as follows:

```
codatatype (sset: 'a) stream =
  SCons (shd: 'a) (stl: "'a stream") (infixr <##> 65)
```

The type is defined via 2 destructors. The first one is `shd`, standing for “stream head”, of type `'a`, representing the first element of the infinite sequence; the second is `stl`, standing for “stream tail”, of type `'a stream`, representing the rest of the sequence when the head is removed. This useful definition does much more than to just declare the 2 destructors. For instance, it also defines the constructor `SCons`, with the notation `##`, relating the the two destructors: `a ## uu` is the infinite sequence with head equal to `a` and tail to `uu`.

In the context of combinatorics on words, the following primitively recursive function (of “HOL-Library.Stream”) is of interest as it prepends a finite word to an infinite word:

```
primrec shift :: "'a list ⇒ 'a stream ⇒ 'a stream" where
  "shift [] s = s"
| "shift (x # xs) s = x ## shift xs s"
```

It is defined on the 2 constructors of `'a list`: the empty word `[]`, and `x # xs`, which stands for the prepending of the letter `a` to the word `xs` (which equals to the concatenation of the singleton `[a]` to the word `xs`). In the formalization, we adopt the notation `·` for the mapping `shift`. Note that it is a bold cdot symbol, compared to the non-bold one for the concatenation of two finite words. This choice is done even though Isabelle allows for overloading of a notation. The reason is that the overloading prevents automatic type inference in some cases, which results in the need to explicitly specify the datatypes of variables often, which in turn makes the claims somewhat less readable.

Prefix and factor of an infinite word are defined in a straightforward manner:

```
definition sprefix :: "'a list ⇒ 'a stream ⇒ bool"
where "sprefix p uu ≡ (∃ vv. uu = p·vv)"

definition sfactor :: "'a list ⇒ 'a stream ⇒ bool"
where "sfactor w uu ≡ (∃ p vv. uu = p·w·vv)"
```

As the names “prefix” and “factor” are already used for prefix and factor of a finite word, the prefix `s` is used in accordance with the library “HOL-Library.Stream” (`hd` is the first letter of a finite word and `tl` is its tail).

The language of an infinite word is just a collection of all its factors.

```
definition word_language ("ℱ")
where "word_language uu ≡ {w. sfactor w uu}"
```

We adopt the notation $\mathcal{F} \text{ uu}$ for the language of an infinite word `uu`. Similarly, the definition of the *factor complexity* of an infinite word, that is, the mapping counting the number of factors of given length, is also straightforward:

```
definition factor_complexity
where "factor_complexity uu ≡ (λi. card ((ℱ uu)i)")"
```

Here, L_i stand for the elements of the language L of length i . The term `card` is the cardinality of a set. The symbol λ introduces a lambda expression, which makes `factor_complexity` a mapping from natural numbers to natural numbers.

As of now, the adaptation of `'a stream` for combinatorics on words was immediate. The next concepts are slightly less obvious. First, we want to produce an infinite word based on an expression for its n -th letter. This is done by the following primitively corecursive function:

```
primcorec snth_stream :: "(nat  $\Rightarrow$  'a)  $\Rightarrow$  'a stream" where
  "shd (snth_stream s) = (s 0)"
| "stl (snth_stream s) = snth_stream ( $\lambda$ i. s (Suc i))"
```

The argument `s` of `snth_stream s` is of type `nat \Rightarrow 'a`, that is, a mapping from natural number to the alphabet, representing the sequence of letters of the infinite word. The corecursive function is defined on the 2 destructors of the infinite word, saying first that its head is `s 0` and its tail is defined using `s` shifted by 1 as `Suc i` is the successor of `i`, i.e., `i+1`.

Another preferred method of producing new infinite words is applying a morphism to some infinite word. This can be achieved by the following combination of existing tools of “HOL-Library.Stream” and the elementary combinatorics on words package:

```
definition morph_stream::"('a list  $\Rightarrow$  'b list)  $\Rightarrow$  'a stream  $\Rightarrow$  'b
stream"
where "morph_stream f uu  $\equiv$  flat (sfilter ( $\lambda$ w. w  $\neq$   $\epsilon$ ) (smap fc
uu))"
```

The mapping `f` is of type `'a list \Rightarrow 'b list` and the definition in fact does not require for it to be a morphism for the same reasons as in the case of a purely morphic language above. The term `fc` stands for the “core” of `f`, which is the underlying mapping of letters (not words) to its images. The mapping `smap` applies `fc` to each element of `uu` producing a sequence of words, more precisely a term of type `'b list list`. To concatenate this sequence of words into the desired word of type `'b list` it suffices to use the mapping `flat`. However, the flattened sequence is first pruned: the empty words in the sequence are ignored using the `sfilter` function. The reason for this step is that the definition of `flat` does not work for empty words in the flattened sequence.

Therefore, the given definition thus results in what we usually mean by a mapping of an infinite word $\mathbf{u} = u_0u_1u_2\dots$:

$$\varphi(\mathbf{u}) = \varphi(u_0)\varphi(u_1)\varphi(u_2)\dots$$

with the marginal exception when $\varphi(\mathbf{u})$ would in fact be a finite word. In that case, we would end up with a term `morph_stream f uu` for which no reasonable claim could be proven.

As one of the motivations to introduce `morph_stream` is to define the fixed point of a morphism, which usually requires some assumption to prevent this case of shrinking an infinite word to a finite one, the definition above is satisfactory.

More specifically, this assumption is that the morphism φ is *prolongeable* on the first letter of its fixed point, that is, we have

$$\lim_{i \rightarrow +\infty} |\varphi^i(u_0)| = +\infty \quad (1)$$

for $\mathbf{u} = u_0u_1u_2\dots$ with $\varphi(\mathbf{u}) = \mathbf{u}$. Such word \mathbf{u} is a *fixed point* of the morphism φ . This concept is formalized as a locale.

```

locale fixed_point = endomorphism +
  fixes u
  assumes
    fixed_point: "(morph_stream f u) = u" and
    shd_grow: "growing f [shd u]"

```

The second assumption stands exactly for the requirement (1). Within this locale, we can indeed show that the language of the fixed point $\mathcal{F} u$ equals the purely morphic language of the morphism f with the axiom equal to the word of length 1 consisting of the first letter of u :

```

lemma wlan_pmor: " $\mathcal{F} u = \text{purely\_morphic\_language } f \text{ ([shd } u \text{])}$ "

```

A purely periodic infinite word, i.e., a word of the form $uuu\dots$, is in fact already present in “HOL-Library.Stream” as `cycle u`, where u is the repetend. We define this notion in a straightforward manner:

```

definition purely_periodic where
  "purely_periodic uu  $\equiv \exists u. uu = \text{cycle } u$ "

```

It is not difficult to prove that the factor complexity of a purely periodic word is eventually constant, and it is bounded by the number of conjugates of the repetend. For this, we use the *primitive root* mapping from `Combinatorics_Words`. Given a non-empty word w , it returns the word $\rho(w)$ which satisfies $w = (\rho(w))^k$ for maximal k .

```

lemma pperiodic_fac_comp: assumes "w  $\neq \varepsilon$ "
  shows "factor_complexity (cycle w) n  $\leq |\varrho w|$ "
  "n  $\geq |\varrho w| \implies \text{factor\_complexity (cycle w) n} = |\varrho w|$ "

```

Analogous definition and similar statements are done for eventually periodic infinite words.

5 Sturmian words

Sturmian words are infinite words that form a very popular subject of study in combinatorics on words. They were studied in the article of 1940 by Morse and Hedlund [15], and they have many equivalent characterizations (c.f [2]). In this formalization, we choose a constructive characterization which allows to parametrize the set of all Sturmian words. More precisely, we choose the concept of lower mechanical words with irrational slope. First, the n -th element needs to be defined for given real parameters α and β :

```

definition lower_mechanical_word_n :: "real  $\Rightarrow$  real  $\Rightarrow$  int  $\Rightarrow$  int"
where
  "lower_mechanical_word_n  $\alpha$   $\beta$  n  $\equiv$  [(n+1)* $\alpha$ + $\beta$ ] - [n* $\alpha$ + $\beta$ ]"

```

The infinite word is then seamlessly produced using `snth_stream` mentioned above.

```

definition lower_mechanical_word :: "real  $\Rightarrow$  real  $\Rightarrow$  int stream"
where
  "lower_mechanical_word  $\alpha$   $\beta$   $\equiv$  snth_stream
  (lower_mechanical_word_n  $\alpha$   $\beta$ )"

```

Real numbers are not part of the main Isabelle package and one must import the theory “HOL.Real”. It provides enough tools to manipulate real numbers to formalize (and prove) the following claims:

```

lemma lmw_rat_per: assumes " $\alpha \in \mathbb{Q}$ "
shows "purely_periodic (lower_mechanical_word  $\alpha$   $\beta$ )"

```

```

lemma lmw_per_rat: assumes "periodic (lower_mechanical_word  $\alpha$ 
 $\beta$ )"
shows " $\alpha \in \mathbb{Q}$ "

```

The claims exhibit the well-known fact that if α is rational, the lower mechanical word with slope α is periodic and vice versa. Hence, if α is irrational, the word `lower_mechanical_word α β` is an aperiodic word which is a Sturmian word.

The proofs of the last two lemmas are done via rotation of the unit circle, which is used in fact in another known characterization of Sturmian words.

6 Structure of the formalization

The formalization presented in the previous 3 sections is available in the Combinatorics on Words Formalized project repository [7] as part of the package “CoW_Infinite” split into 3 theories:

1. `CoW_Infinite/Languages.thy` covers languages in general and (purely) morphic languages, described in [Section 3](#);
2. `CoW_Infinite/Infinite_Words.thy` covers infinite words, described in [Section 4](#);
3. `CoW_Infinite/Sturmian_Words.thy` covers lower mechanical words, described in [Section 5](#).

While the first two theories are intended to serve as a general backbone of the extension of Combinatorics on Words Formalized project to infinite words, the last theory of Sturmian words is rather a proof of concept theory, to demonstrate that the backbone sessions work well on a popular matter.

Acknowledgements

The author acknowledges support by the Czech Science Foundation grant GAČR 20-20621S. The author also expresses his gratitude to the other authors of the Combinatorics on Words Formalized project, Štěpán Holub and Martin Raška.

References

1. Archive of Formal Proofs. <https://www.isa-afp.org/topics.html>
2. Balková, L., Pelantová, E., Starosta, Š.: Sturmian jungle (or garden?) on multiliteral alphabets. *RAIRO-Theor. Inf. Appl.* **44**, 443–470 (2010). <https://doi.org/10.1051/ita/2011002>
3. Berstel, J., Perrin, D.: The origins of combinatorics on words. *Eur. J. Combin.* **28**, 996–1022 (2007). <https://doi.org/http://dx.doi.org/10.1016/j.ejc.2005.07.019>
4. Ehrenfeucht, A., Rozenberg, G.: Repetition of subwords in DOL languages. *Information and Control* **59**(1), 13–35 (1983). [https://doi.org/https://doi.org/10.1016/S0019-9958\(83\)80028-X](https://doi.org/https://doi.org/10.1016/S0019-9958(83)80028-X), <https://www.sciencedirect.com/science/article/pii/S001999588380028X>
5. Holub, Š., Raška, M., Starosta, Š.: Combinatorics on Words Formalized. <https://gitlab.com/formalcow/combinatorics-on-words-formalized>
6. Holub, Š., Raška, M., Starosta, Š.: Binary codes that do not preserve primitivity. In: Blanchette, J., Kovács, L., Pattinson, D. (eds.) *Automated Reasoning*. pp. 369–387. Springer International Publishing, Cham (2022)
7. Holub, Š., Raška, M., Starosta, Š.: Combinatorics on Words Formalized v1.8. <https://gitlab.com/formalcow/combinatorics-on-words-formalized> (2023)
8. Holub, Š., Starosta, Š.: Formalization of Basic Combinatorics on Words. In: Cohen, L., Kaliszyk, C. (eds.) *12th International Conference on Interactive Theorem Proving (ITP 2021)*. Leibniz International Proceedings in Informatics (LIPIcs), vol. 193, pp. 22:1–22:17. Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl, Germany (2021). <https://doi.org/10.4230/LIPIcs.ITP.2021.22>, <https://drops.dagstuhl.de/opus/volltexte/2021/13917>
9. Holub, Š., Raška, M., Starosta, Š.: Combinatorics on Words Basics. *Archive of Formal Proofs* (May 2021), https://isa-afp.org/entries/Combinatorics_Words.html, Formal proof development
10. Holub, Š., Starosta, Š.: Lyndon words formalized in Isabelle/HOL. In: *Developments in Language Theory: 25th International Conference, DLT 2021, Porto, Portugal, August 16–20, 2021, Proceedings*. pp. 217–228. Springer-Verlag, Berlin, Heidelberg (2021), https://doi.org/10.1007/978-3-030-81508-0_18
11. Isabelle generic proof assistant. <https://isabelle.in.tum.de/>
12. Kaliszyk, C., Rabe, F.: A survey of languages for formalizing mathematics. In: Benz Müller, C., Miller, B. (eds.) *Intelligent Computer Mathematics*. pp. 138–156. Springer International Publishing, Cham (2020)
13. Lindenmayer, A.: Mathematical models for cellular interactions in development I. Filaments with one-sided inputs. *Journal of Theoretical Biology* **18**(3), 280–299 (1968). [https://doi.org/https://doi.org/10.1016/0022-5193\(68\)90079-9](https://doi.org/https://doi.org/10.1016/0022-5193(68)90079-9), <https://www.sciencedirect.com/science/article/pii/0022519368900799>

14. Lothaire, M.: *Combinatorics on Words*, Encyclopaedia of Mathematics and its Applications, vol. 17. Addison-Wesley, Reading, Mass. (1983), reprinted in the *Cambridge Mathematical Library*, Cambridge University Press, Cambridge UK, 1997
15. Morse, M., Hedlund, G.A.: Symbolic dynamics II - Sturmian trajectories. *Amer. J. Math.* **62**, 1–42 (1940)
16. Nipkow, T.: A tutorial introduction to structured isar proofs. <https://courses.grainger.illinois.edu/cs576/sp2015/doc/isar-overview.pdf>
17. Nipkow, T., Wenzel, M., Paulson, L.C. (eds.): *Isabelle/HOL*. Springer Berlin Heidelberg (2002)
18. Prouhet, E.: Mémoire sur quelques relations entre les puissances des nombres. *C. R. Acad. Sci. Paris* **33**, 225 (1851)
19. Rozenberg, G., Salomaa, A.: *The book of L.* Springer (1986). <https://doi.org/10.1007/978-3-642-95486-3>
20. Thue, A.: Über unendliche Zeichenreihen. *Skrifter: Matematisk-Naturvidenskapelig Klasse* (1906)
21. Wiedijk, F.: Formalizing 100 theorems. <https://www.cs.ru.nl/~freek/100/>